# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=pcredz
pkgver=100.7bef6cc
pkgrel=1
groups=('blackarch' 'blackarch-scanner')
pkgdesc='A tool that extracts credit card numbers and more from a pcap file or from a live interface.'
arch=('any')
url='https://github.com/lgandx/PCredz'
license=('GPL2')
depends=('python' 'python-python-libpcap' 'libpcap')
makedepends=('git')
source=("$pkgname::git+https://github.com/lgandx/PCredz.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

package() {
  cd $pkgname

  install -Dm 755 Pcredz "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/$pkgname/" Readme.md
}

